Supplementary Material

Mplus Ouput

Figures S1 and S2 show annotated Mplus output for our two power analysis examples. In each, power is identified as the proportion of simulated studies that record a *p*-value of less than or equal to 0.05 for each of the estimated effects. Figure S1 highlights the power estimates for the support provision and support receipt main effects, as well as their interactions with phase. Figure S2 highlights these same effects as well as the contrasts between partner members' analogous effects.

SAS Syntax for Power Analysis of Examples 1 and 2

Figures S3 and S4 provide annotated SAS syntax for simulating, analyzing, and aggregating output for the individual and dyadic power analyses of support's effect on stress described in the manuscript.

Sensitivity Analysis using SAS

Figure S5 provides SAS syntax for conducting the first sensitivity analysis described in the manuscript in which participants and repeated assessments are iterated while keeping the total number of observations equal at each level. The syntax is similar to that presented for the individual power analyses, but the individual power analyses are looped such that power can be estimated for different combinations of parameter specifications.

SPSS Syntax for Power Analysis of Example 1

Figures S6 provides annotated SPSS syntax for simulating, analyzing, and aggregating output for the individual power analysis of support's effect on examinees' stress described in the manuscript.

R Syntax for Power Analysis of Example 1

Figures S7 provides annotated R syntax for simulating, analyzing, and aggregating output for the individual power analysis of support’s effect on examinees’ stress described in the manuscript.

Sensitivity Analysis using R

Figure S8 provides R syntax for conducting the first sensitivity analysis described in the manuscript in which participants and repeated assessments are iterated while keeping the total number of observations equal at each level. The syntax utilizes the function created in Figure S7 and creates another function wrapper to conduct the iterative simulations.

Figure S1. Annotated Mplus output for support receipt and anxiety example.

MODEL RESULTS

POWER

	Population	ESTIMATES Average	Std. Dev.	S. E. Average	M. S. E.	95% Cover	% Sig Coeff
Within Level							
ANX ON							
PHASE	0.400	0.4001	0.0345	0.0347	0.0012	0.947	1.000
PPROV	-0.030	-0.0296	0.0682	0.0706	0.0046	0.953	0.065
PREC	0.170	0.1692	0.0800	0.0785	0.0064	0.940	0.555
REC WITH							
PROV	0.051	0.0509	0.0053	0.0053	0.0000	0.945	1.000
Means							
LANX	0.000	-0.0007	0.0156	0.0150	0.0002	0.943	0.057
PHASE	0.220	0.2200	0.0088	0.0087	0.0001	0.945	1.000
PROV	0.580	0.5806	0.0103	0.0104	0.0001	0.950	1.000
REC	0.560	0.5607	0.0108	0.0106	0.0001	0.939	1.000
PPROV	0.000	0.0001	0.0043	0.0043	0.0000	0.946	0.054
PREC	0.000	0.0000	0.0039	0.0038	0.0000	0.947	0.053
Variances							
LANX	0.500	0.4996	0.0148	0.0150	0.0002	0.949	1.000
PHASE	0.170	0.1699	0.0052	0.0051	0.0000	0.933	1.000
PROV	0.240	0.2398	0.0073	0.0072	0.0001	0.940	1.000
REC	0.250	0.2498	0.0079	0.0075	0.0001	0.939	1.000
PPROV	0.041	0.0409	0.0012	0.0012	0.0000	0.934	1.000
PREC	0.033	0.0330	0.0010	0.0010	0.0000	0.937	1.000
Residual Variances							
ANX	0.420	0.4183	0.0134	0.0134	0.0002	0.938	1.000
Between Level							
Means							
ANX	0.140	0.1387	0.0454	0.0452	0.0021	0.944	0.863
SLOPE1	-0.500	-0.5004	0.0268	0.0271	0.0007	0.961	1.000
SLOPE2	-0.040	-0.0387	0.0287	0.0300	0.0008	0.947	0.234
SLOPE3	0.120	0.1210	0.0403	0.0403	0.0016	0.952	0.844

Figure S2. Annotated Mplus output for dyadic support receipt and anxiety example.

MODEL RESULTS

		Population	ESTIMATES Average	Std. Dev.	S. E. Average	M. S. E.	95% Cover	% Sig Coeff
Within Level								POWER
EANX	ON							
EPHASE		0.400	0.3997	0.0339	0.0337	0.0012	0.950	1.000
EPHPROV		-0.030	-0.0338	0.0698	0.0685	0.0049	0.943	0.086
EPHREC		0.170	0.1685	0.0768	0.0769	0.0059	0.951	0.565
PANX	ON							
PPHASE		0.200	0.2015	0.0352	0.0338	0.0012	0.933	1.000
PPHPROV		-0.030	-0.0327	0.0682	0.0690	0.0047	0.954	0.076
PPHREC		0.040	0.0415	0.0801	0.0766	0.0064	0.934	0.097
EPROV	WITH							
EREC		0.051	0.0508	0.0052	0.0053	0.0000	0.942	1.000
PPROV	WITH							
PREC		0.051	0.0510	0.0056	0.0053	0.0000	0.924	1.000
EANX	WITH							
PANX		0.100	0.1000	0.0100	0.0097	0.0001	0.936	1.000
Residual Variances								
EANX		0.420	0.4187	0.0135	0.0134	0.0002	0.939	1.000
PANX		0.420	0.4182	0.0133	0.0134	0.0002	0.931	1.000
Between Level								
ESLOPE1	WITH							
PSLOPE1		0.007	0.0069	0.0059	0.0058	0.0000	0.939	0.214
ESLOPE2	WITH							
PSLOPE2		0.000	0.0007	0.0032	0.0080	0.0000	1.000	0.000
ESLOPE3	WITH							
PSLOPE3		0.017	0.0162	0.0119	0.0118	0.0001	0.935	0.253
EANX	WITH							
PANX		0.030	0.0289	0.0148	0.0143	0.0002	0.931	0.526
Means								
EANX		0.140	0.1412	0.0464	0.0449	0.0021	0.944	0.872
PANX		0.070	0.0688	0.0459	0.0453	0.0021	0.936	0.326
ESLOPE1		-0.500	-0.4983	0.0268	0.0267	0.0007	0.944	1.000
PSLOPE1		-0.500	-0.4988	0.0271	0.0265	0.0007	0.941	1.000
ESLOPE2		-0.040	-0.0410	0.0282	0.0294	0.0008	0.957	0.276
PSLOPE2		-0.040	-0.0408	0.0269	0.0294	0.0007	0.969	0.259
ESLOPE3		0.120	0.1190	0.0409	0.0395	0.0017	0.930	0.854
PSLOPE3		0.060	0.0613	0.0410	0.0398	0.0017	0.939	0.347

Figure S3. SAS syntax for power analysis of support and anxiety example.

```
*Lane & Hennes JSPR - Example 1;
*POWER ANALYSIS FOR BOLGER ET AL. (2000);

*STATISTICAL MODEL
anx = (B0 + B0i) + (B1 + B1i)*(lanx) + (B2)*phase + (B3 + B3i)*prov) + (B4 + B4i)*rec
                                     + (B5)*prov*phase + (B6)*rec*phase + error
;
DATA EXAMPLE1;
RETAIN SUBNUM;
RETAIN SEED 20160129;
/*estimates for the model effects*/
RETAIN B0 .14; RETAIN B1 -.5; RETAIN B2 .4; RETAIN B3 -.04; RETAIN B4 .12; RETAIN B5 -.03; RETAIN B6 .17;
/*estimates for the random effects*/
RETAIN VSUB_B0 .094; RETAIN VSUB_B1 .022; RETAIN VSUB_B3 .001; RETAIN VSUB_B4 .052;
/*estimate for the error/residual variance*/
RETAIN VRESID .42;
RETAIN SAMPLES 1000;
DO SAMPLE=1 TO SAMPLES;
  DO SUBNUM=1 TO 68;
    B0i=SQRT(VSUB_B0)*RANNOR(SEED); B1i=SQRT(VSUB_B1)*RANNOR(SEED);
    B3i=SQRT(VSUB_B3)*RANNOR(SEED); B4i=SQRT(VSUB_B4)*RANNOR(SEED);
      DO DAY=1 TO 32;
            phase=SQRT(.17)*RANNOR(SEED);
            lanx=SQRT(.5)*RANNOR(SEED);
            prov=SQRT(.24)*RANNOR(SEED);
            rec=SQRT(.25-(.21*.21))*RANNOR(SEED) + SQRT(.21)*prov; /*construct rec variable so that it is
                                                                     correlated with prov .21*/
            anx = (B0 + B0i) + (B1 + B1i)*lanx + (B2)*phase + (B3 + B3i)*prov + (B4 + B4i)*rec +
                  (B5)*prov*phase + (B6)*rec*phase + SQRT(VRESID)*RANNOR(SEED);
OUTPUT;
END; END; END;
DROP SEED b0 b1 b2 b3 b4 b5 b6 vresid vsub_b0 vsub_b1 vsub_b3 vsub_b4 samples;
RUN;

*Check that the variances match what was intended;
PROC MEANS DATA=EXAMPLE1 mean var; var b0i b1i b3i b4i DAY lanx phase prov rec anx; run;

*Turn off all output;
```

```
ODS GRAPHICS OFF; ODS SELECT NONE;

*Analyze the simulated data using the specified model and save the parameter estimates;
PROC MIXED DATA=EXAMPLE1 COVTEST NOCLPRINT;
BY SAMPLE;
CLASS SUBNUM;
MODEL anx = lanx phase prov rec prov*phase rec*phase / S ddf=67,67,67,67,67,67,67;
RANDOM INTERCEPT lanx prov rec / SUBJECT=SUBNUM type=vc;
ODS OUTPUT SOLUTIONF=fixedeffects covparms=randomeffects;
RUN;

*Turn all output back on;
ODS SELECT ALL; ODS GRAPHICS ON; ODS LISTING;

*Collect estimates for the intercept and estimate its power;
data intercept; set fixedeffects; if effect='Intercept';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=intercept; var ESTIMATE sig; run;

*Collect estimates for lagged anxiety and estimate its power;
data lanx; set fixedeffects; if effect='lanx';
if probt le .05 & estimate lt 0 then sig=1; else sig=0; run;
proc means data=lanx; var ESTIMATE sig; run;

*Collect estimates for phase and estimate its power;
data phase; set fixedeffects; if effect='phase';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=phase; var ESTIMATE sig; run;

*Collect estimates for support provision and estimate its power;
data prov; set fixedeffects; if effect='prov';
if probt le .05 & estimate lt 0 then sig=1; else sig=0; run;
proc means data=prov; var ESTIMATE sig; run;

*Collect estimates for support receipt and estimate its power;
data rec; set fixedeffects; if effect='rec';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=rec; var ESTIMATE sig; run;

*Collect estimates for interaction between support provision and phase and estimate its power;
data pprov; set fixedeffects; if effect='phase*prov';
```

```
if probt le .05 & estimate lt 0 then sig=1; else sig=0; run;
proc means data=pprov; var ESTIMATE sig; run;

*Collect estimates for the interaction between support receipt and phase and estimate its power;
data prec; set fixedeffects; if effect='phase*rec';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=prec; var ESTIMATE sig; run;
```

Figure S4. SAS syntax for power analysis of dyadic support and anxiety example.

```
*Lane & Hennes JSPR - Example 2;
*POWER ANALYSIS FOR DYADIC VERSION OF BOLGER ET AL. (2000);

*STATISTICAL MODEL
anx = E*[(B0e + B0ei) + (B1e + B1ei)*(elanx) + (B2e)*ephase + (B3e + B3ei)*eprov) + (B4e + B4ei)*erec
       + (B5e)*eprov*ephase + (B6e)*erec*ephase] +
      P*[(B0p + B0pi) + (B1p + B1pi)*(planx) + (B2p)*pphase + (B3p + B3pi)*pprov) + (B4p + B4pi)*prec
       + (B5p)*pprov*pphase + (B6p)*prec*pphase]
       + error
;
DATA EXAMPLE2;
RETAIN SUBNUM ;
RETAIN SEED 20160129;
/*EXAMINEE*/
/*estimates for the model effects*/
RETAIN B0e .14; RETAIN B1e -.5; RETAIN B2e .4; RETAIN B3e -.04; RETAIN B4e .12; RETAIN B5e -.03; RETAIN B6e
.17;
/*estimates for the random effects*/
RETAIN VSUB_B0e .094; RETAIN VSUB_B1e .022; RETAIN VSUB_B3e .001; RETAIN VSUB_B4e .052;
/*estimate for the residual variance*/
RETAIN VRESIDe .42;

/*PARTNER*/
/*estimates for the model effects*/
RETAIN B0p .07; RETAIN B1p -.5; RETAIN B2p .2; RETAIN B3p -.04; RETAIN B4p .06; RETAIN B5p -.03; RETAIN B6p
.04;
/*estimates for the random effects*/
RETAIN VSUB_B0p .094; RETAIN VSUB_B1p .022; RETAIN VSUB_B3p .001; RETAIN VSUB_B4p .052;
/*estimate for the residual variance*/
RETAIN VRESIDp .42;

/*EXAMINEE-PARTNER Covariances*/
RETAIN COVANX .03;
RETAIN COVLAGANX .007;
RETAIN COVPROV .000;
RETAIN COVREC .017;
RETAIN COVRESID .10;
```

```
RETAIN SAMPLES 1000;
DO SAMPLE=1 TO SAMPLES;
  DO SUBNUM=1 TO 68;
    ANX_DYAD=SQRT(COVANX)*RANNOR(SEED);
    LAGANX_DYAD=SQRT(COVLAGANX)*RANNOR(SEED);
    PROV_DYAD=SQRT(COVPROV)*RANNOR(SEED);
    REC_DYAD=SQRT(COVREC)*RANNOR(SEED);
    RESID_DYAD=SQRT(COVRESID)*RANNOR(SEED);
    DO PARTNER=0 TO 1;
      B0Ei=SQRT(VSUB_B0E)*RANNOR(SEED); B0Pi=SQRT(VSUB_B0P)*RANNOR(SEED);
      B1Ei=SQRT(VSUB_B1E)*RANNOR(SEED); B1Pi=SQRT(VSUB_B1P)*RANNOR(SEED);
      B3Ei=SQRT(VSUB_B3E)*RANNOR(SEED); B3Pi=SQRT(VSUB_B3P)*RANNOR(SEED);
      B4Ei=SQRT(VSUB_B4E)*RANNOR(SEED); B4Pi=SQRT(VSUB_B4P)*RANNOR(SEED);
      DO DAY=1 TO 32;
        phase=SQRT(.17)*RANNOR(SEED);
        lanx=SQRT(.5)*RANNOR(SEED);
        prov=SQRT(.24)*RANNOR(SEED);
        rec=SQRT(.24)*RANNOR(SEED);
        IF PARTNER=0 THEN anx = (B0E + B0Ei + ANX_DYAD) + (B1E + B1Ei + LAGANX_DYAD)*lanx + (B2E)*phase +
                                (B3E + B3Ei + PROV_DYAD)*prov + (B4E + B4Ei + REC_DYAD)*rec +
                                (B5E)*prov*phase + (B6E)*rec*phase +
                                SQRT(VRESIDe)*RANNOR(SEED) + RESID_DYAD;
        IF PARTNER=1 THEN anx = (B0P + B0Pi + ANX_DYAD) + (B1P + B1Pi + LAGANX_DYAD)*lanx + (B2P)*phase +
                                (B3P + B3Pi + PROV_DYAD)*prov + (B4P + B4Pi + REC_DYAD)*rec +
                                (B5P)*prov*phase + (B6P)*rec*phase +
                                SQRT(VRESIDp)*RANNOR(SEED) + RESID_DYAD;
OUTPUT;
END; END; END; END;
DROP SEED samples
covanx covlaganx covprov covrec covresid anx_dyad laganx_dyad prov_dyad rec_dyad resid_dyad
b0e b1e b2e b3e b4e b5e b6e vreside vsub_b0e vsub_b1e vsub_b3e vsub_b4e
b0p b1p b2p b3p b4p b5p b6p vresidp vsub_b0p vsub_b1p vsub_b3p vsub_b4p;
RUN;

*Check that the variances match what was intended;
PROC MEANS DATA=EXAMPLE2 mean var;
var b0ei b1ei b3ei b4ei b0pi b1pi b3pi b4pi DAY lanx phase prov rec anx ; run;

*Create partner dummy variables for analysis;
DATA EXAMPLE2; SET EXAMPLE2;
IF PARTNER=0 THEN E=1; ELSE E=0;
```

```
IF PARTNER=1 THEN P=1; ELSE P=0;
RUN;

*Create covariance matrix to specify which random effects to estimate and constrain to be equal;
data covmatrix;
input parm row col value;
datalines;
1 1 1 1
1 5 5 1
2 2 2 1
2 6 6 1
3 3 3 1
3 7 7 1
4 4 4 1
4 8 8 1
5 1 5 1
6 2 6 1
7 3 7 1
8 4 8 1
9 1 2 0
9 1 3 0
9 1 4 0
9 1 6 0
9 1 7 0
9 1 8 0
9 2 3 0
9 2 4 0
9 2 5 0
9 2 7 0
9 2 8 0
9 3 4 0
9 3 5 0
9 3 6 0
9 3 8 0
9 4 5 0
9 4 6 0
9 4 7 0
9 5 6 0
9 5 7 0
9 5 8 0
9 6 7 0
```

```
9 6 8 0
9 7 8 0
;
run;

data covmatrix; set covmatrix; retain line 0; line=line+1; run;

*replicate the matric for application to each of the samples;
data covmatrix2;
do sample=1 to 1000;
do line=1 to 36;
output;
end; end;
run;
proc sort data=covmatrix2; by line sample; run;
data covmatrix2; merge covmatrix2 covmatrix; by line; run;
proc sort data=covmatrix2; by sample line; run;
data covmatrix2; set covmatrix2; drop line; run;

*Turn off all output;
ODS GRAPHICS OFF; ODS SELECT NONE;

*Analyze the simulate data using the specified model and save the parameter estimates;
PROC MIXED DATA=EXAMPLE2 COVTEST NOCLPRINT;
BY SAMPLE;
CLASS SUBNUM DAY PARTNER;
MODEL anx = E E*lanx E*phase E*prov E*rec E*Prov*phase E*rec*phase
                  P P*lanx P*phase P*prov P*rec P*Prov*phase P*rec*phase
/ S ddf=67,67,67,67,67,67,67 NOINT;
RANDOM E E*lanx E*prov E*rec
         P P*lanx P*prov P*rec     / SUBJECT=SUBNUM type=lin(9) ldata=covmatrix2;
*REPEATED PARTNER / SUBJECT=SUBNUM*DAY type=cs;
ODS OUTPUT SOLUTIONF=fixedeffects covparms=randomeffects;
RUN;

*Turn all output back on;
ODS SELECT ALL; ODS GRAPHICS ON; ODS LISTING;

*Collect estimates for the interaction between support receipt and phase and estimate its power;
data prec; set fixedeffects; if effect='E*phase*rec' | effect='phase*rec*P'; if probt le .05 & estimate gt
0 then sig=1; else sig=0; run;
```

```
proc sort data=prec; by effect; run;
proc means data=prec; by effect; var ESTIMATE sig; run;
```

Figure S5. SAS syntax for sensitivity analysis example.

```
*Lane & Hennes JSPR - Sensitivity Analyses for Example 1;
*POWER ANALYSIS FOR BOLGER ET AL. (2000);

*STATISTICAL MODEL
anx = (B0 + B0i) + (B1 + B1i)*(lanx) + (B2)*phase + (B3 + B3i)*prov) + (B4 + B4i)*rec
                                     + (B5)*prov*phase + (B6)*rec*phase + error
;
*Change # of Participants;
DATA EXAMPLE1;
RETAIN SUBNUM;
RETAIN SEED 20160129;
/*estimates for the model effects*/
RETAIN B0 .14; RETAIN B1 -.5; RETAIN B2 .4; RETAIN B3 -.04; RETAIN B4 .12; RETAIN B5 -.03; RETAIN B6 .17;
/*estimates for the random effects*/
RETAIN VSUB_B0 .094; RETAIN VSUB_B1 .022; RETAIN VSUB_B3 .001; RETAIN VSUB_B4 .052;
/*estimate for the error/residual variance*/
RETAIN VRESID .42;
RETAIN SAMPLES 1000;
ARRAY SIZE[5] (68 102 136 170 204);
DO SSIZE=1 TO 5;
 DO SAMPLE=1 TO SAMPLES;
  DO SUBNUM=1 TO SIZE[SSIZE];
   B0i=SQRT(VSUB_B0)*RANNOR(SEED); B1i=SQRT(VSUB_B1)*RANNOR(SEED);
   B3i=SQRT(VSUB_B3)*RANNOR(SEED); B4i=SQRT(VSUB_B4)*RANNOR(SEED);
      DO DAY=1 TO 32;
            phase=SQRT(.17)*RANNOR(SEED);
            lanx=SQRT(.5)*RANNOR(SEED);
            prov=SQRT(.24)*RANNOR(SEED);
            rec=SQRT(.25-(.21*.21))*RANNOR(SEED) + SQRT(.21)*prov; /*construct rec variable so that it is
                                                                     correlated with prov .21*/
            anx = (B0 + B0i) + (B1 + B1i)*lanx + (B2)*phase + (B3 + B3i)*prov + (B4 + B4i)*rec +
                  (B5)*prov*phase + (B6)*rec*phase + SQRT(VRESID)*RANNOR(SEED);
OUTPUT;
END; END; END; END;
DROP SEED b0 b1 b2 b3 b4 b5 b6 vresid vsub_b0 vsub_b1 vsub_b3 vsub_b4 samples;
RUN;

DATA EXAMPLE1; SET EXAMPLE1;
```

```
if ssize=1 then ssize=size1; else if ssize=2 then ssize=size2; else if ssize=3 then ssize=size3;
else if ssize=4 then ssize=size4; else ssize=size5;
DROP size1 size2 size3 size4 size5;
run;

*Check that the variances match what was intended;
PROC MEANS DATA=EXAMPLE1 mean var; var b0i b1i b3i b4i DAY lanx phase prov rec anx ; run;

*Turn off all output;
ODS GRAPHICS OFF; ODS SELECT NONE;

*Analyze the simulated data using the specified model and save the parameter estimates;
PROC MIXED DATA=EXAMPLE1 COVTEST NOCLPRINT;
BY SSIZE SAMPLE;
CLASS SUBNUM ;
MODEL anx = lanx phase prov rec prov*phase rec*phase / S ddf=67,67,67,67,67,67,67;
RANDOM INTERCEPT lanx prov rec / SUBJECT=SUBNUM type=vc;
ODS OUTPUT SOLUTIONF=fixedeffects covparms=randomeffects;
RUN;

*Turn all output back on;
ODS SELECT ALL; ODS GRAPHICS ON; ODS LISTING;

*Collect estimates for the interaction between support receipt and phase and estimate its power;
data prec; set fixedeffects; if effect='phase*rec';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=prec; by ssize; var ESTIMATE sig; run;

*****************************************************************************************************;
*Change # of Time Points;
DATA EXAMPLE1;
RETAIN SUBNUM;
RETAIN SEED 20160129;
/*estimates for the model effects*/
RETAIN B0 .14; RETAIN B1 -.5; RETAIN B2 .4; RETAIN B3 -.04; RETAIN B4 .12; RETAIN B5 -.03; RETAIN B6 .17;
/*estimates for the random effects*/
RETAIN VSUB_B0 .094; RETAIN VSUB_B1 .022; RETAIN VSUB_B3 .001; RETAIN VSUB_B4 .052;
/*estimate for the error/residual variance*/
RETAIN VRESID .42;
RETAIN SAMPLES 1000;
ARRAY SIZE[5] (32 48 64 80 96);
```

```
DO SSIZE=1 TO 5;
 DO SAMPLE=1 TO SAMPLES;
  DO SUBNUM=1 TO 68;
   B0i=SQRT(VSUB_B0)*RANNOR(SEED); B1i=SQRT(VSUB_B1)*RANNOR(SEED);
   B3i=SQRT(VSUB_B3)*RANNOR(SEED); B4i=SQRT(VSUB_B4)*RANNOR(SEED);
      DO DAY=1 TO SIZE[SSIZE];
            phase=SQRT((7/SIZE[SSIZE])*(1-(7/SIZE[SSIZE])))*RANNOR(SEED);
            lanx=SQRT(.5)*RANNOR(SEED);
            prov=SQRT(.24)*RANNOR(SEED);
            rec=SQRT(.25-(.21*.21))*RANNOR(SEED) + SQRT(.21)*prov; /*construct rec variable so that it is
                                                                      correlated with prov .21*/
            anx = (B0 + B0i) + (B1 + B1i)*lanx + (B2)*phase + (B3 + B3i)*prov + (B4 + B4i)*rec +
                    (B5)*prov*phase + (B6)*rec*phase + SQRT(VRESID)*RANNOR(SEED);
OUTPUT;
END; END; END; END;
DROP SEED b0 b1 b2 b3 b4 b5 b6 vresid vsub_b0 vsub_b1 vsub_b3 vsub_b4 samples ;
RUN;

DATA EXAMPLE1; SET EXAMPLE1;
if ssize=1 then ssize=size1; else if ssize=2 then ssize=size2; else if ssize=3 then ssize=size3;
else if ssize=4 then ssize=size4; else ssize=size5;
DROP size1 size2 size3 size4 size5;
run;

*Check that the variances match what was intended;
PROC MEANS DATA=EXAMPLE1 mean var; var b0i b1i b3i b4i DAY lanx phase prov rec anx ; run;

*Turn off all output;
ODS GRAPHICS OFF; ODS SELECT NONE;

*Analyze the simulated data using the specified model and save the parameter estimates;
PROC MIXED DATA=EXAMPLE1 COVTEST NOCLPRINT;
BY SSIZE SAMPLE;
CLASS SUBNUM ;
MODEL anx = lanx phase prov rec prov*phase rec*phase / S ddf=67,67,67,67,67,67,67;
RANDOM INTERCEPT lanx prov rec / SUBJECT=SUBNUM type=vc;
ODS OUTPUT SOLUTIONF=fixedeffects covparms=randomeffects;
RUN;

*Turn all output back on;
ODS SELECT ALL; ODS GRAPHICS ON; ODS LISTING;
```

```
*Collect estimates for the interaction between support receipt and phase and estimate its power;
data prec; set fixedeffects; if effect='phase*rec';
if probt le .05 & estimate gt 0 then sig=1; else sig=0; run;
proc means data=prec; by ssize; var ESTIMATE sig; run;
```

Figure S6. SPSS syntax for power analysis of support and anxiety example.

```
*Lane & Hennes JSPR – Example 1.
*POWER ANALYSIS FOR BOLGER ET AL. (2000).

SET SEED = 20160129.

*Generate person-level individual differences – Level-2.
INPUT PROGRAM.
LOOP #s=1 to 100.
LOOP #i=1 to 68.
COMPUTE sample=#s.
COMPUTE id=#i.
COMPUTE sub_b0=sqrt(.094)*RV.NORMAL(0,1).
COMPUTE sub_b1=sqrt(.022)*RV.NORMAL(0,1).
COMPUTE sub_b3=sqrt(.001)*RV.NORMAL(0,1).
COMPUTE sub_b4=sqrt(.052)*RV.NORMAL(0,1).
END CASE.
END LOOP.
END LOOP.
END FILE.
END INPUT PROGRAM.
EXECUTE.

DATASET NAME level2.

*Generate time-level responses – Level-1.
INPUT PROGRAM.
LOOP #s=1 to 100.
LOOP #i=1 to 68.
LOOP #t=1 to 32.
COMPUTE sample=#s.
COMPUTE id=#i.
```

```
COMPUTE time=#t.
COMPUTE phase=sqrt(.17)*RV.NORMAL(0,1).
COMPUTE lanx=sqrt(.5)*RV.NORMAL(0,1).
COMPUTE prov=sqrt(.24)*RV.NORMAL(0,1).
COMPUTE rec=sqrt(.25-(.21*.21))*RV.NORMAL(0,1)+sqrt(.21)*prov.
END CASE.
END LOOP.
END LOOP.
END LOOP.
END FILE.
END INPUT PROGRAM.
EXECUTE.

DATASET NAME level1.

*merge the two files.
DATASET ACTIVATE level1.
MATCH FILES /FILE=*
  /TABLE='level2'
  /BY sample id.
EXECUTE.

*calcualte the dependent variable.
COMPUTE anx=(.14+sub_b0)+(-.5+sub_b1)*lanx+(.4)*phase+(-.04+sub_b3)*prov+(.12+sub_b4)*rec
                          +(-.03)*prov*phase+(.17)*rec*phase+sqrt(.42)*RV.NORMAL(0,1).
EXECUTE.

*Split the generated data by sample to be analyzed separately.
SPLIT FILE SEPARATE BY sample.

*NOTE: CANNOT EASILY SAVE THE PARAMETER ESTIMATES FROM MIXED MODELS IN SPSS.
MIXED anx WITH phase lanx prov rec
  /FIXED=phase lanx prov rec phase*prov phase*rec | SSTYPE(3)
```

```
  /METHOD=REML
  /PRINT=SOLUTION TESTCOV
  /RANDOM=INTERCEPT lanx prov rec | SUBJECT(id) COVTYPE(VC).

*ALTERNATIVE: GEE LETS US SAVE THE ESTIMATES - NOTE HOWEVER, THAT THE GEE MODEL IS AN
APPROXIMATION TO THE EXPLICIT MIXED MODEL AND DOES NOT ESTIMATE, AND THUS CANNOT SAVE THE
RANDOM EFFECTS.

*Specify a dataset to save the model estimates.
DATASET DECLARE estimates.

*Run the analyses.
GENLIN anx WITH phase lanx prov rec
  /MODEL phase lanx prov rec phase*prov phase*rec INTERCEPT=YES
 DISTRIBUTION=NORMAL LINK=IDENTITY
  /CRITERIA SCALE=MLE PCONVERGE=1E-006(ABSOLUTE) SINGULAR=1E-012 ANALYSISTYPE=3(WALD)
CILEVEL=95
    LIKELIHOOD=FULL
  /REPEATED SUBJECT=id WITHINSUBJECT=time SORT=YES CORRTYPE=EXCHANGEABLE ADJUSTCORR=YES
COVB=ROBUST
    MAXITERATIONS=100 PCONVERGE=1e-006(ABSOLUTE) UPDATECORR=1
  /MISSING CLASSMISSING=EXCLUDE
  /PRINT CPS DESCRIPTIVES MODELINFO FIT SUMMARY SOLUTION
  /OUTFILE COVB=estimates.

*Select the necessary parameters for estimating power.
DATASET ACTIVATE estimates.
FILTER OFF.
USE ALL.
SELECT IF (ROWTYPE_="EST"  |  ROWTYPE_="SIG").
EXECUTE.

SORT CASES BY sample ROWTYPE_.
```

```
CASESTOVARS
  /ID=sample
  /INDEX=ROWTYPE_
  /GROUPBY=VARIABLE.

*Calculate binary significance for individual samples.
COMPUTE b0_SIG=0.
COMPUTE b1_SIG=0.
COMPUTE b2_SIG=0.
COMPUTE b3_SIG=0.
COMPUTE b4_SIG=0.
COMPUTE b5_SIG=0.
COMPUTE b6_SIG=0.

IF P1.EST>0 & P1.SIG<=.05 b0_SIG=1.
IF P2.EST<0 & P2.SIG<=.05 b1_SIG=1.
IF P3.EST>0 & P3.SIG<=.05 b2_SIG=1.
IF P4.EST<0 & P4.SIG<=.05 b3_SIG=1.
IF P5.EST>0 & P5.SIG<=.05 b4_SIG=1.
IF P6.EST<0 & P6.SIG<=.05 b5_SIG=1.
IF P7.EST>0 & P7.SIG<=.05 b6_SIG=1.
EXECUTE.

*Estimate power – here just the support receipt by phase interaction.
DESCRIPTIVES VARIABLES= P7.EST  b6_SIG
  /STATISTICS=MEAN STDDEV MIN MAX.
```

Figure S7. R syntax for power analysis of support and anxiety example.

```
################################################################################
#packages needed
install.packages("arm")
library(arm)
install.packages("lme4")
library(lme4)
install.packages("lmerTest")
library(lmerTest)

##########################################################################################################
# power analysis

#generate data
Ex1.fake <- function (J,K){
  time <- rep(seq(1,K,length=K),J) # K measurements per person
  person <- rep(1:J, each=K)       # J person IDs
  phase <- rnorm(J*K,0,.17)          # time level phase
  lanx <- rnorm(J*K,0,.5)          # time level lag anxiety
  prov <- rnorm(J*K,0,.24)          # time level provision
  rec  <- sqrt(.25)*((1-(.21*.21))*rnorm(J*K,0,1)+sqrt(.21)*prov) # time level receipt
  #fixed effects
  b0 <-  .14    # true intercept value
  b1 <- -.50    # true lag anxiety estimate
  b2 <-  .40    # true phase estimate
  b3 <- -.04    # true provision estimate
  b4 <-  .12    # true receipt estimate
  b5 <- -.03    # true phase * provision estimate
  b6 <-  .17    # true phase * receipt estimate
  #random effects
  vsub.b0 <- .094    # true between person variance in the intercept (i.e. anxiety)
  vsub.b1 <- .022    # true between person variance in the lag anxiety slope
```

```
  vsub.b4 <- .052    # true between person variance in the receipt slope
  vresid <-  .420    # true within person variance in anxiety
  #combine fixed and random effects per person
  b0.int    <- rnorm(J,b0,sqrt(vsub.b0))  #generate an intercept for each person
  b1.laganx <- rnorm(J,b1,sqrt(vsub.b1))  #generate a slope for lag anxiety for each person
  b4.rec    <- rnorm(J,b4,sqrt(vsub.b4))  #generate a slope for receipt for each person
  anx <- rnorm(J*K, b0.int[person]            #use the person's intercept
                  +b1.laganx[person]*lanx   #lag anxiety slope
                  +b2*phase                 #average phase effect
                  +b3*prov                  #average provision effect
                  +b4.rec[person]*rec       #average plus person-specific receipt effect
                  +b5*phase*prov            #average phase by provision effect
                  +b6*phase*rec             #average phase by receipt effect
                  ,sqrt(vresid))            #residual
  return(data.frame(person,time,phase,prov,rec,lanx,anx))
}

data <- Ex1.fake(68,32) #generate an example dataset to make sure it worked
lme.power <- lmer(anx ~ (1 | person) + (-1 + lanx | person) + (-1 + rec | person)
                + lanx + phase + prov + rec + prov*phase + rec*phase,
                data=data)
summary(lme.power)

# this function loops the data generation and analyzes it
Ex1.power <- function (J,K,n.sims=1000){      #default to 1000 simulations if not specified
  signif <- rep(NA,n.sims)                    # a vector that wwill record if the effect of interest is sig
  for (s in 1:n.sims){
    fake <- Ex1.fake(J,K)                     #generate a fake dataset
    lme.power <- lmer(anx ~ (1 | person) + (-1 + lanx | person) + (-1 + rec | person)
                  + lanx + phase + prov + rec + prov*phase + rec*phase, data=fake) #analyze it
    est <- fixef(lme.power)["rec"]      #save the parameter estimate
    se <- se.fixef(lme.power)["rec"]    #save the standard error
    signif[s] <- (abs(est)-2*se)>0      #calculate significance - returns TRUE/FALSE
```

```
  }
  power <- mean(signif)
  return(power)
}

Ex1.power(J=68,K=32,n.sims=1000)
```

Figure S8. R syntax for sensitivity analysis example.

```
## create a wrapper to incorporate multiple scenarios (i.e. sensitivity analyses) ##

## Specify sample sizes to examine ##
Jvals <- c(68, 102, 136, 170, 204)
Kvals <- c(32, 48, 64, 80, 96)
sample.sizes <- expand.grid(Jvals, Kvals)
names(sample.sizes) <- c("J", "K")
## To hold results ##
powvals <- rep(NA, nrow(sample.sizes))
## Calculate power for each combination of J, K ##
for(i in 1:nrow(sample.sizes)){
  ## Get the next set of sample sizes and calculate power ##
  tmpss <- as.numeric(sample.sizes[i,])
  tmppow <- Ex1.power(tmpss[1], tmpss[2], 50)
  powvals[i] <- tmppow
}

powvals

library(ggplot2)
ggplot(sample.sizes, aes(x=J, y=powvals)) + geom_line() + facet_wrap(~ K) +
  xlab("Participant Sample Size") + ylab("Power")
```